Final Project
IEOR 160

Ben LeRoy, Abbey Chaver, Emily Chou, Jackie Ko

April 28, 2015

1 Executive Summary

To come up with the most profit-maximizing plan for the new building project at UC Berkeley, CalInvestor has hired BenBey-Em Kompany (BBEK) to come up with a project portfolio for scheduling over three years.

Given limited resources and project scheduling constraints, BBEK modeled the cases with integer programming in AMPL to determine an efficient portfolio. In accordance with CalInvestors management, a case was evaluated without preemption, meaning a project that is running must continue without pause until it is completed. However, with many years of experience in consulting, BBEK suggested that CalInvestor consider scheduling with preemption, since it can boost profits significantly. Furthermore, CalInvestor was considering filing for an increased budget, and asked BBEK to provide cost-profit analysis for a higher monthly budget.

BBEK ran four cases for CalInvestor, involving variations in evaluations with vs. without preemption and original budget ($5 million) vs. new budget ($7 million). With careful analysis, BBEK determined that it is best for CalInvestor to maintain its current practices and plan– that is, scheduling without preemption and with original monthly budget of $5 million. Scheduling with preemption did result a 1.34% increase in profit, but that is not significant enough to be worth the complication in management that preemption introduces. By raising the budget to $7 million, CalInvestor would result in a small 1.34% increase in profit. By evaluating the fourth case with preemption and new budget, CalInvestor would result in an 2.3% in profit, which is $2.4 million. Having access to additional funding totaling $72 million ($2 mil · 36 months) to make a mere $2.4 million in additional total profit is not very cost-efficient, therefore not desirable.

2 Introduction and Background

2.1 Overview

CalInvestor has been appointed by UC Berkeley to finance the new building for campus expansion. Optimizing the schedule for the multi-million dollar building requires completing dozens of projects. With predetermined constraints on budget and resources, CalInvestor has hired BBEK to help select and schedule a project portfolio in order to maximize profit over three years.

Each month, CalInvestor is granted $5 million for project operational costs. Due to government regulations, no more than three projects can be started in the same month (However, they can have more than three projects running simultaneously). With these resource constraints and limitations, CalInvestor identified twelve projects to be selected for the project portfolio. The projects operational costs, duration, and revenue are given in the tables below:

Operational Cost (Table 1):

Project	1	2	3	4	5	6	7	8	9	10	11	12
c_i ($ mil/month)	1.1	1.2	1.3	1.4	2.5	3.6	1.7	1.8	1.9	2	1.8	1.9

Duration and Revenue (Table 2):

Project	1	2	3	4	5	6	7	8	9	10	11	12
d_i (months)	7	6	10	9	8	6	7	12	10	12	12	12
b_i ($ millions)	2	8	15	4	10	11	16	23	20	50	60	70

2.2 Technological Constraints

Launching of some of the more innovative projects requires the completion of several more fundamental projects.

a. Project 3 must be completed before project 9 can start.

b. Project 1 and 2 must be completed before project 10 can start.

c. Project 1 must be completed before project 11 can start.

d. Project 2 must be completed before project 12 can start.

2.3 Resource Constraints

Some resources and risk limitations mandate that some projects cannot run at the same time.

a. Projects 1, 3, and 8 requires the use of confidential data, which requires all analyses being run on the centralized computer lab at UC Berkeley. Therefore, at most one of these projects can be running at any given time.

b. Projects 2 and 4 require collaboration with partner companies in Wall Street. Due to the limited human resources in NY, these two projects cannot overlap.

c. Project 5 and 6 are both expensive. Hence, to lower the financial risk, there will be a gap of at least one month between operating the two projects.

2.4 Project Completion Constraints

Projects 10, 11 and 12 require the completion of prior projects in order to be carried out. In particular:

a. Project 10 can begin when either project 4 or 5 is finished.

b. Project 11 can begin when either project 6, 7, or 8 is finished.

c. Project 12 can begin when either project 9 or 11 is finished.

2.5 Condition Variables

1. **Preemption**

   For the ease of project management, CalInvestor requests BBEK to schedule the project portfolio with no preemption. That is, once a project is launched, the project will be implemented till fully completed. However, through many contracting experiences, BBEK recommends considering scheduling with preemption since it may boost profits significantly. BBEK will present profit analysis of both cases to CalInvestor for comparison.

2. **Budget Variation**

   There is a possibility that CalInvestor can receive more project funding (to an increased total amount of $7 millions/month) from the UC system. To persuade the finance department, CalInvestor asks BBEK to provide a profit analysis with a new budget limit of $7 million/month, with and without preemption.

In the end, BBEK will run four models, with 2 conditions to test, with subvariables (a) representing original situation

1. Preemption
   (a) Without preemption (original condition in Part 1)
   (b) With preemption
2. Budget (per month)
   (a) Original budget of $5 million
   (b) New budget of $7 million

	Without Preemption (Original)	With Preemption
Budget of $5 Million (Original)	Case 1: Condition 1a, 2a	Case 2: Condition 1b, 2a
Budget of $7 Million	Case 3: Condition 1a, 2b	Case 4: Condition 1b, 2b

2.6 Approach to problem

To determine an efficient portfolio, BBEK modeled the situation with integer programming in AMPL with heavy use of binary variables. In accordance with CalInvestors management, the code does not allow for preemption, meaning a project that is running must continue without pause until it is completed. Although it is simpler to manage projects in one burst, allowing preemption can increase profits so the code is tweaked in the next section to show the difference this condition can make. Furthermore, to show the impact of increasing CalInvestors monthly budget, the code is re-run with a different number. The code was also run with and without the preemption requirement.

3 Modeling

3.1 Parameters

We have some newly created parameters, giving more flexibility to BBEK if CalInvestors wanted to run models changing these up.

months For a more general formulation, we defined a parameter "months" that gives the total number of months we can use for a certain portfolio. We decide to use months as our time unit because each project's smallest unit of time is in months. For this portfolio we are given 3 years, so the total number of months is 36. In the code we use j to run through $\{1, 2, 3, ..., months\}$.

n Similarly, our code contains a parameter n that represents the total number of projects able to be run in the portfolio. In the current portfolio, we have a total of 12 possible projects, so $n = 12$. In the code we usually index through the different projects by using i through $\{1, 2, 3, ..., n\}$.

monthly_cost In our code we have the parameter *monthly_cost* which is a holder variable for the companys monthly budget constraint. We coded this as a parameter since we'd like to be able to vary the monthly budget constraint between \$5 million and \$7 million dollars a month.

As noted in the **Introduction and Background** section, we also coded in certain provided information:

$\mathbf{c_i}$ This denotes the monthly operating cost for project i (in millions of dollars), $\forall i \in \{1, ..., n\}$

$\mathbf{d_i}$ This denotes the number of months it takes to complete project i (duration of project), $\forall i \in \{1, ..., n\}$

$\mathbf{b_i}$ This denotes the revenue gained when project i is completed (in millions of dollars), $\forall i \in \{1, ..., n\}$

3.2 Variables

We used 3 variables, defined below.

$\mathbf{x_{i,j}}$ is a binary variable that is defined to be: $\begin{cases} 1 & \text{if the project } i \text{ starts in time } j \\ 0 & \text{otherwise} \end{cases}$

$\mathbf{y_{i,j}}$ is a binary variable that is defined to be: $\begin{cases} 1 & \text{if the project } i \text{ is operating at time } j \\ 0 & \text{otherwise} \end{cases}$

$\mathbf{v_{i,j}}$ is a binary variable that is defined to be: $\begin{cases} 1 & \text{if in the } j \text{ month the project has finished/ or is finishing} \\ 0 & \text{otherwise} \end{cases}$

3.3 Objective Function

Maximize Profit:

$$\sum_{\mathbf{i=1}}^{\mathbf{n}} \sum_{\mathbf{j=1}}^{\text{months}} (\mathbf{b_i} - \mathbf{d_i} \cdot \mathbf{c_i}) \cdot \mathbf{x_{i,j}}$$

3.4 Logical Constraints (Part I)

Our initial constraints served to define our variables properly.

1. Each project can only begin once:

$$\sum_{j=1}^{\text{months}} x_{i,j} \leq 1 \qquad \forall i \in \{1, ..., n\} \tag{1}$$

2. If the project starts, it must finish.

$$\sum_{j=1}^{\text{months}} y_{i,j} \geq d_i \cdot \sum_{j=1}^{\text{months}} x_{i,j} \qquad \forall i \in \{1, ..., n\} \tag{2}$$

3. Only run a project in months during and after the the initialization occurs:

$$d_i \cdot \sum_{j=1}^{m} x_{i,j} \geq \sum_{j=1}^{m} y_{i,j} \qquad \forall i \in \{1, ..., n\}, \forall m \in \{1, ..., \text{months}\} \tag{3}$$

4. To make $\mathbf{v_{i,j}}$ work correctly: turn on $\mathbf{v_{i,j}}$ if and only if project i has been completed by month j:

$$\cdot \sum_{j=1}^{m} y_{i,j} \geq d_i \cdot v_{i,m} \qquad \forall i \in \{1, ..., n\}, \forall j \in \{1, ..., \text{months}\} \tag{4}$$

$$\frac{1}{d_i} \cdot (\sum_{j=1}^{m} (y_{i,j}) - d_i + 1) \leq v_{i,m} + .999 \qquad \forall i \in \{1, ..., n\}, \forall j \in \{1, ..., \text{months}\} \tag{5}$$

The second equation uses .999 like a M value, to only force V to be 1 once the project uses the number of months it needs to finish the project.

3.5 Logic Constraints (Part II)

Our client posed several scheduling constraints which we represented as linear inequalities.

5. (Logic requirement 4) $\sim$ Completing certain projects before others start:

(a) Completing project 3 before 9 is started:

$$x_{9,j+1} \leq v_{3,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{6}$$

$$x_{9,1} = 0 \tag{7}$$

(b) Complete 1 and 2 before completing 10 is started:

$$x_{10,j+1} \leq v_{1,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{8}$$

$$x_{10,j+1} \leq v_{2,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{9}$$

$$x_{10,1} = 0 \tag{10}$$

(c) Complete 1 before project 11 is started:

$$x_{11,j+1} \leq v_{1,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{11}$$

$$x_{11,1} = 0 \tag{12}$$

(d) Complete 2 before project 12 is started:

$$x_{12,j+1} \leq v_{2,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{13}$$

$$x_{12,1} = 0 \tag{14}$$

6. (Logic requirement 5) $\sim$ Running only one type of project at once (and more):

(a) Run only a single project out of 1, 3 and 8:

$$y_{1,j} + y_{3,j} + y_{8,j} \leq 1 \qquad \forall j \in \{1, ..., \text{months}\} \tag{15}$$

(b) Run only a single project out of 2 and 4:

$$y_{2,j} + y_{4,j} \leq 1 \qquad \forall j \in \{1, ..., \text{months}\} \tag{16}$$

(c) Projects 5 and 6 cannot occur at the same time, and there must be at least a one-month gap between them:

$$y_{5,j} + y_{6,j+1} \leq 1 \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{17}$$

$$y_{5,j} + y_{6,j} \leq 1 \qquad \forall j \in \{1, ..., \text{months}\} \tag{18}$$

$$y_{5,j} + y_{6,j-1} \leq 1 \qquad \forall j \in \{2, ..., \text{months}\} \tag{19}$$

7. (Logic requirement 6) ∼ Completing one out of several prerequisite projects:

(a) Do not begin 10 unless 4 and/or 5 have been completed:

$$x_{10,j+1} \leq v_{4,j} + v_{5,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{20}$$

(b) Do not begin 11 unless 5, 6, and/or 7 have been completed:

$$x_{11,j+1} \leq v_{5,j} + v_{6,j} + v_{7,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{21}$$

(b) Do not begin 12 unless 9 and/or 11 have been completed:

$$x_{11,j+1} \leq v_{9,j} + v_{11,j} \qquad \forall j \in \{1, ..., \text{months} - 1\} \tag{22}$$

8. (Logic requirement 7) ∼ No more than three projects can begin in one month:

$$\sum_{i=1}^{n} x_{i,j} \leq 3 \qquad \forall j \in \{1, ..., \text{months}\} \tag{23}$$

9. (Logic requirement 8) ∼ Do not spend more than 5 million in any one month:

$$\sum_{i=1}^{n} c_i \cdot x_{i,j} \leq \mathit{monthly\_cost} \qquad \forall j \in \{1, ..., \text{months}\} \tag{24}$$

Note: *monthly_cost* = \$5 million originally, \$ 7 million in Case 3 and Case 4.

10. (Logic requirement 9) ∼ No preemption: once begun, a project must run to completion without gaps:

$$d_i \cdot x_{i,m} \leq \sum_{j=m}^{m+d_i-1} y_{i,j} \qquad \forall i \in 1, ..., n, \forall m \in \{1, ..., (\text{months} - d_i)\} \tag{25}$$

3.6 Variations

To run the model with preemption, we simply removed Logic Constraint 9.

To increase the monthly budget to 7 million dollars, we simply replaced constraint 8, by changing *monthly_cost* to \$7 million.

4 Results and Analysis

4.1 Case 1:

Case 1 is evaluated with a monthly budget of \$5 million and no preemption. In other words, once a project is launched, the project will be implemented till fully completed. A total of 8 projects are started and completed, costing \$120.8 million and producing a net profit of \$104.2 million. The project scheduling diagram for this case is shown below:



4.2 Case 2:

Case 2 is evaluated with a monthly budget of \$5 million and with preemption. A total of 9 projects are started and completed, costing \$142.4 million and producing a profit of \$105.6 million. The project scheduling diagram for this case is shown below:



Compared to Case 1, Case 2 is evaluated with preemption (a project can be paused and restarted). Preemption allows a more flexible allocation of resources, resulting in an extra completed project and hence more profit.

4.3 Case 3:

Case 3 is evaluated with a new monthly budget of \$7 million and without preemption. A total of 9 projects are started and completed, costing \$142.4 million and producing a profit of \$105.6 million. The project scheduling diagram for this case is shown below:



Compared to Case 1, Case 3 is evaluated with an increase in monthly budget from \$5 million to \$7 million, which is 40% more than the original. This allows more projects to be started and operated in the same month, resulting in an increase of number of completed projects and hence profit. For example, project 3 can be started earlier so it runs with projects 4 and 11 .

4.4 Case 4:

Case 4 is evaluated with a new monthly budget of $7 million and with preemption. A total of 10 projects are started and completed, costing $161.4 million and producing a profit of $106.6 million. The project scheduling diagram for this case is shown below:



Compared to Case 1, Case 4 is evaluated with an increase in budget and preemption. This allows for both increase in budget and flexibility in resource allocation. As a result, two more projects are completed and there is $2.2 million gain in profit.

5 Recommendations and Conclusion

Allowing preemption does not increase profit significantly, and makes managing projects more complicated since every ongoing project must be monitored each month to check for preemption. With preemtion you loose efficiencies of consistenly running projects (as mentioned before), which is not taken into account in the project reports. Increasing the budget without preemption made an equally insignificant difference in profit. If preemption and budget were both changed then there would be a $2.4 million increase in profit, but that is only 2.3% more than the original. On the next page there is a summary chart of the differences in profits depending on which case is used.

Cases	1	2	3	4
Total Profit ($ millions)	104.2	105.6	105.6	106.6
% Increase from original case 1	N/A	1.34	1.34	2.30

Based on the models, the effort required to change current activities (described in Case 1) is not worth the benefits. Since the $2 million must be taken from somewhere else from the UC system, the finance department may decide the overall cost not worth the slight profit boost. Furthermore, preemption would also have to be implemented with the budget change, which complicates project management significantly since every projects status would need to be checked monthly; without preemption, projects that start will simply run until completion without much managerial supervision. Preemption often delays project completion as well, and since the company makes no profit until a project is fully finished, CalInvestor may not be able to afford the heavier initial costs.

In conclusion, to maximize profit, BBEK recommends that CalInvestor schedule the projects without preemption and with a monthly budget of $5 million according to the following scheduling diagram (Case 1):



5.1 Additional Recommendations

Some recommendations BBEK suggests CalInvestor to consider includes:

- Relaxation in constraints:
  - Constraint 5b: CalInvestor can seek different partners not necessarily based in Wall Street, New York. By relaxing this constraint, CalInvestor can have more flexibility in project assignment and possibly finish more projects.
- Data on financial risk:
  - Since some projects have higher costs than others, projects can be ranked in order of preference for completing.
  - For example, since projects 9 to 12 are innovative, there is a higher chance of revenue not meeting expectations due to technological failures, so they would be lower priority for scheduling
- Extensions:
  - Though the budget and preemption were changed for the entire duration for each case, they could be varied for certain projects only, to check if profits would increase.
  - For instance, if budget was increased when running the most expensive projects 5 and 6, CalInvestor would not need to change its policies permanently but still be capable of running more projects simultaneously.

#Appendix:
#Model Code

```
param n; #number of projects (12)
param months; #number of months (=36)
param monthly_cost;

param c {1..n}; # operational cost in millions/month for project j
param d {1..n}; # number of months required to complete j
param b {1..n}; # revenue upon completion of project j


var x {1..n, 1..months} binary; # indicates whether project i is
initiated in month j
var y {1..n, 1..months} binary; # indicates whether project i is
running in month j
var v {1..n, 1..months} binary; # indicates whether project i is
completed by month j

maximize Profit:
        # sum of (revenue - duration*cost) for all projects that are
done
        sum {i in 1..n, j in 1..months} (b[i] - d[i]*c[i])*x[i,j];



subject to

# Project can only begin once
defX {i in 1..n}: sum{j in 1..months} x[i,j] <= 1;

# Must complete project in 3 years if it is begun
completion {i in 1..n}: sum{j in 1..months} y[i, j] >= d[i] * sum{j in
1..months} x[i,j];

# Running months occur only after beginning project
defY {i in 1..n, m in 1..months}: d[i]*sum{j in 1..m}x[i,j] >= sum{j
in 1..m}y[i,j];

# V turns on iff project has finished by month m
defV {i in 1..n, m in 1..months}: sum{j in 1..m} y[i, j] >=
d[i]*v[i,m];
defV2 {i in 1..n, m in 1..months}: (1/d[i])*sum{j in 1..m} y[i,j] <=
(v[i,m] + 0.999);


#4
# Do not start 9 unless 3 is completed, with edge case
r4a1 {j in 1..(months-1)}: x[9, j+1] <= v[3, j];
```

```
r4a2: x[9, 1] = 0;

# Do not start 9 unless 1 and 2 are completed, with edge case
r4b1 {j in 1..(months-1)}: x[10, j+1] <= v[1, j];
r4b2 {j in 1..(months-1)}: x[10, j+1] <= v[2, j];
r4b3: x[10, 1] = 0;

# Do not start 11 unless 1 is completed, with edge case
r4c1 {j in 1..(months-1)}: x[11, j+1] <= v[1, j];
r4c2: x[11, 1] = 0;


# Do not start 12 unless 2 is completed, with edge case
r4d1 {j in 1..(months-1)}: x[12, j+1] <= v[2, j];
r4d2: x[12, 1] = 0;


#5
# only one of project 1, 3, and 8 running at a time
r5a {j in 1..months}: y[1, j] + y[3, j] + y[8, j] <= 1;

#only one of project 2 and 4 running at a time
r5b {j in 1..months}: y[2, j] + y[4, j] <= 1;

#schedule gap between of one month between project 5 and 6
r5c1 {j in 1..(months-1)}: y[6, j+1] + y[5, j] <= 1;
r5c2 {j in 1..months}: y[6, j] + y[5, j] <= 1;
r5c3 {j in 2..months}: y[6, j-1] + y[5, j] <= 1;


#6
# Do not start 10 unless 4 and/or 5 is completed
r6a {j in 1..(months-1)}: x[10, j + 1] <= v[4, j] + v[5, j];

# Do not start 11 unless 5, 6 and/or 7 is completed
r6b {j in 1..(months-1)}: x[11, j + 1] <= v[5, j] + v[6, j] + v[7, j];

# Do not start 12 unless 9 and/or 11 is completed
r6c {j in 1..(months-1)}: x[12, j + 1] <= v[9, j] + v[11, j];


#7
# No more than three projects can begin in one month
r7 {j in 1..months}: sum{i in 1..n}x[i,j] <= 3;


#8
# Do not spend more than 5 million in any month (or 7 later)
r8 {j in 1..months}: sum{i in 1..n}(c[i]*y[i,j]) <= monthly_cost;
```

```
#9
# No preemption allowed
# Might have trouble referencing d[i] properly for each case # nope -
all good
r9 {i in 1..n, m in 1..(months - d[i])}: d[i]*x[i, m] <= sum{j in m..
(m + d[i] - 1)}y[i, j];
```

```
#Appendix:
#data file

param n:=12;
param months:=36;

param monthly_cost:=5; #change to 7 for Part 3

param c :=
1       1.1
2       1.2
3       1.3
4       1.4
5       2.5
6       3.6
7       1.7
8       1.8
9       1.9
10      2
11      1.8
12      1.9
;
param d :=
1       7
2       6
3       10
4       9
5       8
6       6
7       7
8       12
9       10
10      12
11      12
12      12
;

param b :=
1       2
2       8
3       15
4       4
5       10
6       11
7       16
8       23
9       20
10      50
11      60
12      70
```

;